APPENDIX

COMPUTER SOFTWARE

In this Appendix we simply list the computer programs from IONCAP that are used for the various noise calculations. The programs are given in the following order: SUBROUTINE ANOIS1, which calculates the 1 MHz atmospheric noise levels for two adjacent four-hour time blocks by calling SUBROUTINE NOISY; SUBROUTINE NOISY which uses the Fourier coefficients to compute the 1 MHz atmospheric noise value; SUBROUTINE GENFAM, which computes the atmospheric noise level at the desired frequency and the values of D_μ, D_ℓ, $\sigma_{F_{am}}$, σ_{D_μ}, and σ_{D_ℓ}; the new SUBROUTINE GENOIS, which computes the sum noise process and its statistics; and the old GENOIS for comparison. Additional information is contained in the various comment statements and the body of this report.

```
      SUBROUTINE ANOIS1
C
C     THIS ROUTINE DETERMINES THE 1 MHZ ATMOSPHERIC NOISE
C
C     FOURIER SERIES IN LATITUDE AND LONGITUDE FOR TWO DISCRETE
C     LOCAL TIME BLOCKS
C
      COMMON/ANOIS/ATNU,ATNY,CC,TM,XEFF,RCNSE,DU,DL,SIGM,SIGU,SIGL,KJ,JK
      COMMON /CON /D2R, DCL, GAMA, PI, PI2, PIO2, R2D, RZ, VOFL
      COMMON / DON / ALATD, AMIN, AMIND, DLONG, DMP, ERTR,
     1  PMP, PWR, RLAT, RLATD, RLONG, RLONGD, RSN, SIGTR,
     2 TLAT, TLATD, TLONG, TLONGD, FLUX, SSN, ATMNO, D90R, D50R,
     3 D10R, D90S, D50S, D10S
      COMMON /FILES/ LUO,LUI,LU25,LU26
      COMMON / TIME / IT, GMT, UTIME(24), GMTR, XLMT(24),ITIM,JTX
      COMMON / TWO / F2D(16,6,6), DUD(5,12,5),FAM(14,12),
     A  SYS(9,16,6), PERR(9,4,6), P(29,16,8),ABP(2,9)
C  LMT AT RCVR SITE
      IF(F2D(1,1,1)) 90, 90, 100
C.....NO IONOSPHERIC LONG TERM DATA BASE FILE
C.....SET NOISE TO ZERO HERE (-204 IN SUBROUTINE GENOIS)
C.....THE USER CAN INPUT ANY VALUE AS MAN-MADE NOISE (RESET IN GENOIS)
   90 ATNU = 0.0
      ATNY = 0.0
      RETURN
  100 CC = GMTR
      KJ= 6
      IF(CC-22.) 105,110,110
  105 KJ = CC/4. +1.
  110 TM = 4*KJ-2
      IF(CC-TM) 115,120,125
  115 JK = KJ -1
      GO TO 130
  120 JK = KJ
      GO TO 130
  125 JK = KJ+1
  130 IF(JK) 135,135,140
  135 JK =6
      GO TO 150
  140 IF(JK-6) 150,150,145
  145 JK = 1
C.....EAST LONGITUDE (IN DEGREES)
  150 CEG= RLONGD
  165 XLA = RLAT * R2D
C.....LATITUDE (IN DEGREES) "+" IS NORTH
      CALL NOISY(KJ,XLA,CEG,ATNU)
      CALL NOISY(JK,XLA,CEG,ATNY)
      RETURN
      END
```

```
      SUBROUTINE NOISY (KJ, XLA, CEG, ANOS)
C     NOISY IS A GENERAL PURPOSE ROUTINE USED TO EVALUATE A FOURIER
C     SERIES IN TWO VARIABLES.
C     KJ --- NUMBER OF FOURIER COEFFICIENT ARRAY TO BE USED
C     XLA --- GEOGRAPHIC LATITUDE, DEGREES,
C     CEG --- GEOGRAPHIC EAST LONGITUDE, DEGREES
C     ANOS --- NOISE VALUE, MEDIAN POWER DB ABOVE KTB
C     ABP --- NORMALIZING FACTORS FOR FOURIER SERIES
C     KJ = 1 TO 6 IS ATMOSPHERIC NOISE, KJ = 7 IS LAND MASS MAP AND
C     KJ = 8 IS RATIO OF F2 HEIGHT OF MAXIMUM TO SEMITHICKNESS
C
C     * NOTE - XLA, CEG, ANOS, ABP ARE NOT ALWAYS AS PREVIOUSLY DEFINED
C     FOURIER VARIABLES AND ATMOSPHERIC RADIO NOISE
C
      COMMON / TWO / F2D(16,6,6), DUD(5,12,5),FAM(14,12),
     A SYS(9,16,6), PERR(9,4,6),P(29,16,8),ABP(2,9)
      COMMON /SWTCH/ INIL,OSSN,OMONTH,OTIME,ODIP,OLAT,OLONG
      DIMENSION SX (15), SY(29), ZZ (29)
      IF (KJ - 8)105, 100, 105
C.....LIMITS OF FOURIER SERIES
 100  LM = 15
      LN = 10
      GO TO 110
C.....LIMITS OF FOURIER SERIES
 105  LM = 29
      LN = 15
C.....HALF ANGLE (IN RADIANS)
 110  Q = .0087266466 * CEG
      IF(CEG .EQ. OLONG .AND. INIL .EQ. 0) GO TO 118
C.....LONGITUDE SINES
      DO 115 K = 1, 15
 115  SX(K)=SIN(Q*K)
      OLONG=CEG
 118  CONTINUE
C.....LONGITUDE SERIES
      DO 125 J = 1, LM
      R = 0.
      DO 120 K = 1, LN
 120  R = R + SX (K) * P (J, K, KJ)
 125  ZZ (J) = R + P (J, 16, KJ)
C.....ANGLE PLUS 90 DEGREES (IN RADIANS)
      Q = .01745329252 * (XLA + 90.)
      IF(XLA .EQ. OLAT .AND. INIL .EQ. 0) GO TO 145
C.....LATITUDE SERIES
      DO 140 J=1,29
 140  SY(J)=SIN(Q*J)
      INIL=0
      OLAT=XLA
 145  CONTINUE
      R = 0.
      DO 130 K = 1, LM
 130  R = R + SY (K) * ZZ (K)
C.....FINAL FOURIER SERIES EVALUATION (NOTE LINEAR NORMALIZATION)
 135  ANOS = R + ABP(1,KJ)+ABP(2,KJ)* Q
      RETURN
      END
```

```
      SUBROUTINE GENFAM(Y2,IBLK,FREQ,Z,FA,DU,DL,DMS,DUS,DLS)
C
C     GENFAM CALCULATES THE FREQUENCY DEPENDENCE OF THE ATMOSPHERIC
C     NOISE AND GETS DECILES AND PREDICTION ERRORS FROM TABLES
C
      COMMON / TWO / F2D(16,6,6), DUD(5,12,5),FAM(14,12),
     A  SYS(9,16,6), PERR(9,4,6), P(29,16,8),ABP(2,9)
      DIMENSION V(5)
      IF(F2D(1,1,1) ) 90,90,95
C.....NO IONOSPHERIC LONG TERM DATA BASE FILE (SET IN SUBROUTINE GENOIS)
   90 FA = 0.0
      DU=9.
      DL=7.
      DUS=1.5
      DLS=1.5
      DMS=3.
      RETURN
   95 CONTINUE
      IBK=IBLK
C.....CHECK IF LATITUDE IS NORTH OR SOUTH
      IF (Y2)100, 105, 105
  100 IBK = IBK + 6
  105 U1 = - .75
      X = .43429 * ALOG (FREQ)
      U = (8. * 2. * * X - 11.) / 4.
      KOP = 1
  110 PZ = U1 * FAM (1, IBK) + FAM (2, IBK)
      PX = U1 * FAM (8, IBK) + FAM (9, IBK)
      DO 115 I = 3, 7
      PZ = U1 * PZ + FAM (I, IBK)
  115 PX = U1 * PX + FAM (I + 7, IBK)
      IF(KOP-1) 120,120,125
  120 CZ = Z * PZ + PX
      CZ = Z + Z - CZ
      U1 = U
      KOP = 2
      GO TO 110
  125 FA = CZ * PZ + PX
      DO 145 I = 1, 5
      Y = DUD (1, IBK, I)
      DO 140 J = 2, 5
      IF (J - 5)140, 130, 140
  130 IF (X - 1.)140, 140, 135
  135 X = 1.
  140 Y = Y * X + DUD (J, IBK, I)
  145 V (I) = Y
      DU = V (1)
      DL = V (2)
      DUS = V (3)
      DLS = V (4)
      DMS = V (5)
      RETURN
      END
```

```
      SUBROUTINE GENOIS
C
C     THIS ROUTINE COMPUTES THE COMBINED NOISE DISTRIBUTION
C
      COMMON / DON / ALATD, AMIN, AMIND, DLONG, DMP, ERTR,
     1  PMP, PWR, RLAT, RLATD, RLONG, RLONGD, RSN, SIGTR,
     2 TLAT, TLATD, TLONG, TLONGD, FLUX, SSN, ATMNO, D90R, D50R,
     3 D10R, D90S, D50S, D10S
      COMMON/FILES/ LUO,LUI,LU25,LU26
      COMMON/ANOIS/ATNU,ATNY,CC,TM,XEFF,RCNSE,DU,DL,SIGM,SIGU,SIGL,KJ,JK
      COMMON /TON /ADJ, ADS, ATMO, GNOS, GOT, PWRDB, ZCNSE, REL, SL, SLS
     1, SPR, SU, SUS, TIMER, XADJN, ZEFF, XNOISE, XTLOS, ZNOISE, NF
      COMMON/FRQ/FREL(29),FREQ,JMODE
      COMMON / ION / IANT(3,2), NTR(2), IEA, IFQB, IFQE, IGRAPH, IHRE,
     A IHRO, IHRS, JO, LUFP, METHOD, MONPR, NDAY, NES, NOISE, NPAT,
     B NPSL, NRSP, NUMO
      COMMON / METSET /  ITRUN, ITOUT, JTRUN(40), JTOUT(40)
      COMMON/RON/RAT(5),CLCK(5),ABIY(5),ARTIC(5),SIGPAT(5),EPSPAT(5),
     A FI(3,5),YI(3,5),HI(3,5),FX(3,5),HPRIM(30,3),HTRUE(30,3),
     B FVERT(30,3),KFX,AFAC(30,3),HNOR(3),HTR(50),FNSQ(50)
      COMMON /RTANT /XETA, XSIG, XEPS, XND, XNL, XNH, TEX (4), ITANT, IR
     1 ANT, RETA, RSIG, REPS, RND, RNL, RNH, REX(4), TEFF, REFF, KASANT
      COMMON / TWO / F2D(16,6,6), DUD(5,12,5),FAM(14,12),
     A  SYS(9,16,6), PERR(9,4,6), P(29,16,8),ABP(2,9)
      COMMON / ZON / ABPS(7), CREL(7), EFF(7), FLDST(7), GRLOS(7),
     1HN (7), HP (7), PROB (7), RELY (7), RGAIN (7), SIGPOW (7), SN (7),
     2 SPRO (7), TGAIN (7), TIMED (7), TLOSS (7), B (7), FSLOS (7), ADV
     C (7),OBF(7),NMODE(7),NPROB,NREL,TLLOW(7),TLHGH(7)
      DIMENSION XNINT(4)
C.....MAN-MADE NOISE LEVELS AS GIVEN BY CCIR REPORT 258.
      DATA XNINT /76.8, 72.5, 67.2, 53.6/
      DATA DFAC,BFAC,CFAC/7.87384,30.99872,5.56765/
C
C     7.87384=SQRT(2*1.282**2*4.34294**2)
C
C     30.99872=(1.282**2)(4.34294**2)
C
C     5.56765=4.34294*1.282
C
C.....DATA IS FA VALUES AT 1 MHZ
C.....ATNU, ATNY ARE DB .GT. KTB FOR 1 MHZ
C.....ATNZ, ATNX ARE DB .GT. KTB FOR DESIRED FREQ,DUM
C.....ATNOS, GNOS, XNOIS ARE DB .GT. KTB FOR ALL CALCULATIONS
C.....AND ARE CONVERTED TO DBW(1 HZ BWDTH) AT END OF ROUTINE
C.....UPPER LIMIT IS 55 MHZ FOR NOISE
       DUME = AMIN1(FREQ,55.)
      MAN=NOISE
C  FREQUENCY DEPENDENCE ATMOSPHERIC NOISE
      IF(F2D(1,1,1)) 85, 90, 90
C.....NO IONOSPHERIC LONG TERM DATA BASE FILE
C.....FORCE MAN-MADE NOISE OR GALACTIC NOISE
   85 ATNOS = 0.
```

```
      DUA=9.
      DLA=7.
      SMA = 3.
      SUA = 1.5
      SLA = 1.5
      GO TO 95
   90 CONTINUE
C.....FREQUENCY DEPENDENCE
      CALL GENFAM(RLAT,KJ,DUME,ATNU,ATNZ,DU,DL,SIGM,SIGU,SIGL)
      CALL GENFAM(RLAT,JK,DUME,ATNY,ATNX,DX,DQ,SIGZ,SIGX,SIGSQ)
C.....BEGIN OF INTERPOLATION ON LOCAL TIME
      SLOP = ABS(CC-TM)/4.
      ATNOS = ATNZ + (ATNX - ATNZ) * SLOP
      DUA= DU +(DX-DU)*SLOP
      DLA= DL +(DQ-DL)*SLOP
        SMA= SIGM+ (SIGZ-SIGM)*SLOP
        SUA= SIGU +(SIGX-SIGU)*SLOP
        SLA= SIGL+(SIGSQ-SIGL)* SLOP
C
C     (DUA/DFAC)**2=(DUA/1.282)**2/(2*4.34294**2)
C                  =(DUA/SQRT(2*1.282**2*4.34294**2))**2
C                  =(DUA/7.87384)**2
C
   95 AU=EXP((DUA/DFAC)**2 + (ATNOS/4.34294))
      VU=AU*AU*(EXP(DUA*DUA/BFAC)-1.)
      AL=EXP((DLA/DFAC)**2 + (ATNOS/4.34294))
C
C     DLA*DLA/BFAC=(DLA/1.282)**2/(4.34294)**2
C                 =DLA**2/30.99872
C
      VL=AL*AL*(EXP(DLA*DLA/BFAC)-1.)
C  GALACTIC NOISE
      IF(FREQ - FI(3,KFX)) 100, 100, 105
C.....GALACTIC NOISE DOES NOT PENETRATE
  100 GNOS = 0.
      GO TO 110
  105 GNOS = 52. - 23. * ALOG10(FREQ)
  110 DUG=2.
      AT=EXP((DUG/DFAC)**2 + (GNOS/4.34294))
      AU=AU+AT
      VU=VU+AT*AT*(EXP(DUG*DUG/BFAC)-1.)
      DLG=2.
      AT=EXP((DLG/DFAC)**2 + (GNOS/4.34294))
      AL=AL+AT
      VL=VL+AT*AT*(EXP(DLG*DLG/BFAC)-1.)
      SMG = .5
      SUG = .2
      SLG = .2
C  MAN MADE  NOISE
       MAN=NOISE
      XNOIS = MAN
      MA = IABS(MAN)
      ZNOISE=XNOIS
      IF(MAN) 120, 114, 115
```

```
C.....INDICATES -164 ON USER INPUT
  114 MA = 4
      GO TO 120
C.....CONVERT 3 MHZ DB .LT. 1 WATT INPUT VALUE TO FA AT 1 MHZ
  115 XNOIS=204.0-XNOIS+13.22
C.....OBTAIN FA AT DESIRED FREQUENCY
      XNOIS = XNOIS - 27.7 * ALOG10(FREQ)
      GO TO 125
C....NEGATIVE ON USER INPUT INDICATES INDEX
  120 MA = MINO(4,MA)
      CONN=27.7
      IF(MA .EQ. 4) CONN=28.6
      XNOIS = XNINT(MA) - CONN * ALOG10(FREQ)
      ZNOISE = 204.0 - XNINT(MA) + 13.22
  125 DUM=9.7
      AT=EXP((DUM/DFAC)**2+(XNOIS/4.34294))
      AU=AU+AT
      VU=VU+AT*AT*(EXP(DUM*DUM/BFAC)-1.)
      DLM=6.
      AT=EXP((DLM/DFAC)**2+(XNOIS/4.34294))
      AL=AL+AT
      VL=VL+At*AT*(EXP(DLM*DLM/BFAC)-1.)
      SUM=1.5
      SMM=5.4
      SLM=1.5
C.....RECEIVER ANTENNA EFFICIENCY
      CALL GAIN(2,KASANT,0.0,FREQ,GDUM,REFF)
      XEFF = REFF
      ZEFF=XEFF
C.....SET ARRAY FOR ALL POSSIBLE MODES
      DO 196 IM=1,6
  196 EFF(IM) = XEFF
C.....NOW DETERMINATION OF NOISE LEVEL IS ITS-78(HFMUFES4)
C.....SWITCH TO DB .GT. WATT
      ATNOS=ATNOS-204.
      GNOS=GNOS-204.
      XNOIS=XNOIS-204.
      SIGTSQ=ALOG(1.+VU/(AU*AU))
      XRNSE= 4.34294*(ALOG(AU)-SIGTSQ/2.) -204.
C.....UPPER DECILE
C
C     CFAC=4.34294*1.282
C         =5.56765
C
      DU= CFAC*SQRT(SIGTSQ)
      SIGTSQ=ALOG(1.+VL/(AL*AL))
C.....LOWER DECILE
      DL= CFAC*SQRT(SIGTSQ)
      IF(ITRUN - 8) 205, 210, 205
  205 QPA = 10. ** ((ATNOS - XRNSE) * .1)
      QPG = 10.**((GNOS -XRNSE)*.1)
C.....PREDICTION ERRORS
C.....SIGM IS MEDIAN, SIGU IS UPPER AND SIGL IS LOWER
      QPM = 10.**((XNOIS-XRNSE)*.1)
```

```
      SIGM= SQRT((QPA*SMA)**2 +(QPG*SMG)**2 +(QPM*SMM)**2)
C
C     0.23026=1.0/4.34294
C
      PV=QPA*EXP((DUA-DU)*.23026)
      SIGU= (PV*SUA)**2+((PV-QPA)*SMA)**2
      PV=QPG*EXP((DUG-DU)*.23026)
      SIGU=SIGU+(PV*SUG)**2+((PV-QPG)*SMG)**2
      PV=QPM*EXP((DUM-DU)*.23026)
      SIGU=SQRT(SIGU+(PV*SUM)**2+((PV-QPM)*SMM)**2)
      PV=QPA*EXP((DLA-DL)*.23026)
      SIGL= (PV*SLA)**2+((PV-QPA)*SMA)**2
      PV=QPG*EXP((DLG-DL)*.23026)
      SIGL=SIGL+(PV*SLG)**2+((PV-QPG)*SMG)**2
      PV=QPM*EXP((DLM-DL)*.23026)
      SIGL=SQRT(SIGL+(PV*SLM)**2+((PV-QPM)*SMM)**2)
C RCVR SITE NOISE = TOTAL NOISE + ANTENNA EFFICENCY (ADDED TO SIGNAL
C   WITH GAIN)
  210 RCNSE = XRNSE + XEFF
      ZCNSE=RCNSE
      ATMNO=ATNOS
      XADJN=1.
      XNOISE=XNOIS
      ATMO=ATNOS
      RETURN
      END
```

```
      SUBROUTINE GENOIS
C
C     THIS ROUTINE COMPUTES THE COMBINED NOISE DISTRIBUTION
C
      COMMON / DON / ALATD, AMIN, AMIND, DLONG, DMP, ERTR,
     1  PMP, PWR, RLAT, RLATD, RLONG, RLONGD, RSN, SIGTR,
     2 TLAT, TLATD, TLONG, TLONGD, FLUX, SSN, ATMNO, D90R, D50R,
     3 D10R, D90S, D50S, D10S
      COMMON/FILES/LUO,LUI,LU25,LU26
      COMMON/ANOIS/ATNU,ATNY,CC,TM,XEFF,RCNSE,DU,DL,SIGM,SIGU,SIGL,KJ,JK
      COMMON /TON /ADJ, ADS, ATMO, GNOS, GOT, PWRDB, ZCNSE, REL, SL, SLS
     1, SPR, SU, SUS, TIMER, XADJN, ZEFF, XNOISE, XTLOS, ZNOISE, NF
      COMMON/FRQ/FREL(29),FREQ,JMODE
      COMMON / ION / IANT(3,2), NTR(2), IEA, IFQB, IFQE, IGRAPH, IHRE,
     A IHRO, IHRS, JO, LUFP, METHOD, MONPR, NDAY, NES, NOISE, NPAT,
     B NPSL, NRSP, NUMO
      COMMON / METSET /  ITRUN, ITOUT, JTRUN(40), JTOUT(40)
      COMMON/RON/RAT(5),CLCK(5),ABIY(5),ARTIC(5),SIGPAT(5),EPSPAT(5),
     A FI(3,5),YI(3,5),HI(3,5),FX(3,5),HPRIM(30,3),HTRUE(30,3),
     B FVERT(30,3),KFX,AFAC(30,3),HNOR(3),HTR(50),FNSQ(50)
      COMMON /RTANT /XETA, XSIG, XEPS, XND, XNL, XNH, TEX (4), ITANT, IR
     1 ANT, RETA, RSIG, REPS, RND, RNL, RNH, REX(4), TEFF, REFF, KASANT
      COMMON / TWO / F2D(16,6,6), DUD(5,12,5),FAM(14,12),
     A  SYS(9,16,6), PERR(9,4,6), P(29,16,8),ABP(2,9)
      COMMON / ZON / ABPS(7), CREL(7), EFF(7), FLDST(7), GRLOS(7),
     1HN (7), HP (7), PROB (7), RELY (7), RGAIN (7), SIGPOW (7), SN (7),
     2 SPRO (7), TGAIN (7), TIMED (7), TLOSS (7), B (7), FSLOS (7), ADV
     C (7),OBF(7),NMODE(7),NPROB,NREL,TLLOW(7),TLHGH(7)
      DIMENSION XNINT(4)
C.....MAN-MADE NOISE LEVELS
      DATA XNINT /125., 136., 148., 164./
C.....BUT COMBINATION IS NOT
C.....CALCULATION OF NOISE LEVEL IS ITSA-1
C.....ATNU, ATNY ARE DB .GT. KTB FOR 1 MHZ
C.....ATNZ, ATNX ARE DB .GT. KTB FOR FREQ
C.....ATNOS, GNOS, XNOIS ARE DB .LT. 1 WATT IN 1 HZ BAND AT FREQ
C.....UPPER LIMIT IS 55 MHZ FOR NOISE
       DUME = AMIN1(FREQ,55.)
      MAN=NOISE
C  FREQUENCY DEPENDENCE ATMOSPHERIC NOISE
      IF(F2D(1,1,1)) 85, 90, 90
C.....NO IONOSPHERIC LONG TERM DATA BASE FILE
C.....FORCE MAN-MADE NOISE OR GALACTIC NOISE
   85 ATNOS = 204.
      DUA = 9.
      DLA = 7.
      SMA = 3.
      SUA = 1.5
      SLA = 1.5
      GO TO 95
   90 CONTINUE
```

```
C.....FREQUENCY DEPENDENCE
      CALL GENFAM(RLAT,KJ,DUME,ATNU,ATNZ,DU,DL,SIGM,SIGU,SIGL)
      CALL GENFAM(RLAT,JK,DUME,ATNY,ATNX,DX,DQ,SIGZ,SIGX,SIGSQ)
C.....BEGIN OF INTERPOLATION ON LOCAL TIME
      SLOP = ABS(CC-TM)/4.
      ATNOS = - (ATNZ + (ATNX - ATNZ) * SLOP) + 204.
      DUA= DU +(DX-DU)*SLOP
      DLA= DL +(DQ-DL)*SLOP
        SMA= SIGM+ (SIGZ-SIGM)*SLOP
        SUA= SIGU +(SIGX-SIGU)*SLOP
        SLA= SIGL+(SIGSQ-SIGL)* SLOP
C.....END OF INTERPOLATION ON LOCAL TIME
C  GALACTIC NOISE
   95 IF(FREQ - FI(3,KFX)) 100, 100, 105
C.....GALACTIC NOISE DOES NOT PENETRATE
  100 GNOS = 204.
      GO TO 110
  105 GNOS = 165. + 9.555 * ALOG(FREQ / 3.)
  110 DUG = 2.
      DLG =2.
      SMG = .5
      SUG = .2
      SLG = .2
C  MAN MADE  NOISE
       MAN=NOISE
      XNOIS = MAN
      MA = IABS(MAN)
      ZNOISE=XNOIS
      IF(MAN) 120, 114, 115
C.....INDICATES -164 ON USER INPUT
  114 MA = 4
      GO TO 120
C.....ACTUAL VALUE IF POSITIVE ON USER INPUT
  115 XNOIS = XNOIS + 12.160 * ALOG(FREQ / 3.)
      MA= -MAN
      GO TO 125
C.....NEGATIVE ON USER INPUT INDICATES INDEX
  120 MA = MINO(4,MA)
      XNOIS = XNINT(MA) + 12.160 * ALOG(FREQ/3.)
      ZNOISE = XNINT(MA)
      MA= -XNINT(MA)
  125 DUM =9.
      DLM =7.
      SUM=1.5
      SMM=3.
      SLM=1.5
C.....RECEIVER ANTENNA EFFICIENCY
      CALL GAIN(2,KASANT,0.0,FREQ,GDUM,REFF)
      XEFF = REFF
      ZEFF=XEFF
C.....SET ARRAY FOR ALL POSSIBLE MODES
      DO 196 IM=1,6
  196 EFF(IM) = XEFF
```

```
C.....NOW DETERMINATION OF NOISE LEVEL IS ITS-78(HFMUFES4)
C.....SWITCH TO DB .GT. WATT
      ATNOS = - ATNOS
      GNOS = - GNOS
      XNOIS = - XNOIS
C ADD THE NOISES (RANDOM PHASE APROXIMATION=ADD THE POWER IN WATTS)
C.....MEDIAN
      XRNSE= 4.343*ALOG((10.**(ATNOS*.1)) + (10.**(GNOS*.1))
     A +(10.**(XNOIS*.1)))
C  CALCULATE THE DECILES AND VARIANCE BY EQ. 37, P. 29 OF THE THEORY OF
C  ERROR BY YARDLEY BEERS, MCGRAW HILL.
C
C.....UPPER DECILE
      DU= ABS(4.343*ALOG(10.**((ATNOS+DUA)*.1)  + 10.**((GNOS+DUG)*.1)
     A +10.**((XNOIS+DUM)*.1))- XRNSE)
C.....LOWER DECILE
      DL= ABS( 4.343 *ALOG(10.**((ATNOS+DLA)*.1) +10.**((GNOS+DLG)*.1)
     A +10.**((XNOIS+DLM)*.1)) -XRNSE)
      IF(ITRUN - 8) 205, 210, 205
  205 QPA = 10. ** ((ATNOS - XRNSE) * .1)
      QPG = 10.**((GNOS -XRNSE)*.1)
C.....PREDICTION ERRORS
C.....SIGM IS MEDIAN, SIGU IS UPPER AND SIGL IS LOWER
      QPM = 10.**((XNOIS-XRNSE)*.1)
      SIGM= SQRT((QPA*SMA)**2 +(QPG*SMG)**2 +(QPM*SMM)**2)
      SIGU= SQRT((DUA*SUA*QPA**2/DU)**2 +(DUG*SUG*QPG**2/DU)**2
     A +(DUM*SUM*QPM**2/DU)**2)
      SIGL = SQRT((DLA*SLA*QPA**2/DL)**2 +(DLG*SLG*QPG**2/DL)**2
     A + (DLM * SLM * QPM ** 2 / DL) ** 2)
C RCVR SITE NOISE = TOTAL NOISE + ANTENNA EFFICENCY (ADDED TO SIGNAL
C   WITH GAIN)
  210 RCNSE = XRNSE + XEFF
      ZCNSE=RCNSE
      ATMNO=ATNOS
      XADJN=1.
      XNOISE=XNOIS
      ATMO=ATNOS
      RETURN
      END
```